Pure Trade Equilibrium Models

- By choice of preference structure, the pure trade model was equivalent to a system of equations – complementarity was not really an issue
  - That is, the equations could have been defined as equalities, and a solution consistent with an equilibrium would result
  - However, if preferences that can result in zero demands for some goods for a consumer, then complementarity would matter
  - What type of preferences would result in this?

Equilibrium Models

  - This is not always the case as there may be
    - Policy inequalitites such as quotas
    - Participation limits
    - Etc.
  - Models with production may also be stated as systems of equations (complementarity doesn't matter) with some kinds of production technology
    - What kind of technology will do this?

Equilibrium Models with Production

- What kind of production technology results in complementarity conditions that matter?
  - Technology with less than full substitution possibilities (e.g., Leontief technology)
- With production included, we have more agents whose behavior needs description
  - How shall we describe producer behavior?

Equilibrium Models with Production

- Producers – profit maximizers
  - Technology options:
    - Single output/multiple input
    - Single input/multiple output
    - Multiple inputs and outputs with input/output separability
    - Leontief
  - First and last are the most common in empirical models (GTAP uses the first)

Equilibrium Models: Production (cont'd.)

- Single output/multiple input technology: $y = f(x) = f(x_1, x_2, \ldots, x_n)$
  - Producer behavior is described as:

$$\underset{x}{\text{maximize}}\ pf(x) - w'x = pf(x_1, \ldots, x_n) - \sum_{i=1}^{n} w_i x_i$$

  - which has first-order conditions:

$$-p\frac{\partial f(x)}{\partial x_i} + w_i \geq 0,\ \ x_i\left[-p\frac{\partial f(x)}{\partial x_i} + w_i\right] = 0$$

Equilibrium Models: Production (cont'd.)

- Note that in writing down the FOC's the problem has again been converted to a minimization problem – this is due to the conventions for writing an MCP for GAMS
- Also notice that if the technology displays constant returns to scale (CRTS), the level of production may be indeterminate – however, the use of inputs per unit of output will be determined in this case

Equilibrium Models: Production (cont'd.)

- With CRTS, an alternative way to write the problem is in terms of a per unit production function as follows:

$$\underset{x}{\text{maximize}}\,(pf(x) - w'x)y = y\left[pf(x_1,\ldots,x_n) - \sum_{i=1}^{n} w_i x_i\right]$$

- which has FOC's:

$$-p\frac{\partial f(x)}{\partial x_i}y + w_i y \geq 0,\ x_i\left[-p\frac{\partial f(x)}{\partial x_i}y + w_i y\right] = 0,$$

$$-pf(x) + w'x \geq 0,\ y[-pf(x) + w'x] = 0$$

Equilibrium Models: Production (cont'd.)

- The first of these conditions says that the marginal value product for an input cannot exceed its cost
- The second says that if the marginal value product of an input is less than its cost, it will not be used
- The third says that profits cannot be positive (CRTS)
- And the fourth says that the level of production can be positive only if profits are zero (and that if profits are negative, production will be zero)

Equilibrium Models: Production (cont'd.)

- Leontief technology: expressed in terms of a "netput vector", *a*, which has positive components corresponding to outputs and negative components corresponding to inputs
  - Producer behavior is described by:

$$\underset{y}{\text{maximize}}\; p'ay = \sum_{i=1}^{n} p_i a_i y$$

  - with FOC's: $-p'a \geq 0,\; y(-p'a) = 0$

Equilibrium Models: Production (cont'd.)

- The first FOC says that profits cannot be positive (CRTS)
- The second FOC says that the level of production can be positive only if profits are zero (and that if profits are negative, production will be zero)

Equilibrium Models: Production (cont'd.)

- Typically, production will be described by many Leontief activities, and these are usually assembled into a matrix, *A*, with each column corresponding to a Leontief activity
  - The FOC's for the entire productive sector of the economy are then expressed jointly as:

    $$-p'A \geq 0, -p'Ay = 0$$

  - Or, no productive activity can make positive profits, and an activity can produce output only if its profits are zero

Equilibrium Models: Production (cont'd.)

- To illustrate, consider an example
  - Another two consumer (Dick and Jane) model
  - Production is specified as Leontief

Equilibrium Example with Production

- Two consumers: Dick and Jane
- Two goods: food and clothes
- Two factors of production: Land (Dick owns 20 units of land and Jane owns 10) and Labor (Dick owns 5 units of labor and Jane owns 20)

GE with Production

- Cobb-Douglas preference parameters for Dick are 0.75 for food and 0.25 for clothes, and for Jane are 0.60 for food and 0.40 for clothes
- Factors of production are not directly consumed
- There are four possible technologies for producing food and clothes – these are Leonteif, multiple input/multiple output technologies

GE with Production (cont'd.)

- Production technology is specified through the following four activities (positives for outputs and negatives for inputs)

tech #	1	2	3	4
food	10	0	5	6
clothes	0	5	3	2.5
land	-1	-1	-1	-1
labor	-.5	-2	-1.2	-1.3

GE with Production (cont'd.)

- Markets are as they were before except that
  - Endowments are only for factors,
  - Demands are only for consumer goods
- Agent first-order conditions are unchanged for consumers, but producer first-order conditions are added
- Note that complementarity matters for this problem

GE with Production (cont'd.)

- As usual, the first thing we will want is to introduce some sets (note use of subsets):

```
SETS
  cons       Consumers               / dick,jane /
  io         Inputs and outputs
                                     /food,clothes,land,labor/
  good(io)   Consumer goods          / food,clothes /
  fact(io)   Factors of production   / land,labor /
  method     Production methods      / a1*a4 / ;
ALIAS (good,g) ;
```


GE with Production (cont'd.)

- Next, specify our data

```
TABLE endow(cons,io) Factor endowments
          land      labor
 dick     20        5
 jane     10        20 ;

TABLE alpha(cons,io) Cobb-Douglas value shares
          food      clothes
 dick     0.75      0.25
 jane     0.60      0.40 ;
```

GE with Production (cont'd.)

- TABLE a(io,method) Input-output technology

	a1	a2	a3	a4
food	10	0	5	6
clothes	0	5	3	2.5
land	-1	-1	-1	-1
labor	-.5	-2	-1.2	-1.3 ;

- (Notice that here is the payoff to defining the set io and the subsets for goods and factors)

GE with Production (cont'd.)

- Next, define the problem variables (and bounds)
  - POSITIVE VARIABLES
  - p(io) Prices of goods and factors
  - x(io,cons) Demands for goods
  - y(method) Level of use of a the technology
  - lambda(cons) Marginal utility of wealth ;
  - x.lo(io,cons) = 0.001 ;
  - p.fx('food') = 1 ;
- Note that "food" is the numeraire here

GE with Production (cont'd.)

- Now, declare the equations:

```
EQUATIONS
 market(io)          Demand cannot exceed supply
 foc(good,cons)      FOC's for demand
 loss(method)        Loss (-profit) level for method
 budget(cons)        Consumer budget constraints ;
```


GE with Production (cont'd.)

- Now define the structure of the equations:

```
market(io)    ..
   sum(cons,endow(cons,io))$fact(io) +
   sum(method,a(io,method)*y(method))
   - sum(cons,x(io,cons))$good(io) =g= 0 ;
```

- For each good, the sources are endowments (factors) and output from production (goods), and the uses are as inputs to production (factors) or for consumption (goods)

GE with Production (cont'd.)

- Consumer behavior (as before)

```
foc(good,cons)..
   -prod(g,x(g,cons)**alpha(cons,g))
   *alpha(cons,good)/x(good,cons)
   + lambda(cons)*p(good) =g= 0 ;
```


GE with Production (cont'd.)

- Producer behavior (profits not positive)

```
loss(method)  ..
   -sum(io,p(io)*a(io,method)) =g= 0 ;
```

- (Expressing the technology matrix in terms of netput vectors simplifies these relationships as well as the "market" conditions or supply/demand relationships)

GE with Production (cont'd.)

- Consumer budget constraints
  - budget(cons) ..
  -    -sum(good,p(good)*x(good,cons))
  -    + sum(fact,p(fact)*endow(cons,fact)) =g= 0 ;
- (Expenditures are only for consumption goods, while income comes from ownership of factors)

GE with Production (cont'd.)

- The model and solve statements are similar to the pure trade case:
  - MODEL production
  -    / market.p, foc.x, loss.y, budget.lambda / ;
  - SOLVE production USING MCP ;

GE with Production (cont'd.)

- Model output is similar to the pure trade case:

---- EQU market Demand cannot exceed supply

	LOWER	LEVEL	UPPER	MARGINAL
food	.	.	+INF	1.000
clothes	.	.	+INF	2.492
land	-30.000	-30.000	+INF	8.231
labor	-25.000	-25.000	+INF	3.537

GE with Production (cont'd.)

---- EQU foc First order conditions on demand

	LOWER	LEVEL	UPPER	MARGINAL
food .dick	.	.	+INF	136.735
food .jane	.	.	+INF	91.837
clothes.dick	.	.	+INF	18.289
clothes.jane	.	.	+INF	24.568

GE with Production (cont'd.)

- ---- EQU loss Loss (-profit) level for method

	LOWER	LEVEL	UPPER	MARGINAL
a1	.	.	+INF	15.714
a2	.	2.846	+INF	.
a3	.	.	+INF	14.286
a4	.	0.600	+INF	.

- Notice that activities "a2" and "a4" make negative profits and are not used

GE with Production (cont'd.)

- ---- EQU budget Consumer budget constraints

	LOWER	LEVEL	UPPER	MARGINAL
dick	.	.	+INF	0.454
jane	.	.	+INF	0.354

GE with Production (cont'd.)

---- VAR p Prices of goods and factors

	LOWER	LEVEL	UPPER	MARGINAL
food	1.000	1.000	1.000	2.558E-13
clothes	.	2.492	+INF	.
land	.	8.231	+INF	.
labor	.	3.537	+INF	.

GE with Production (cont'd.)

---- VAR x Demands for goods

	LOWER	LEVEL	UPPER	MARGINAL
food .dick	0.001	136.735	+INF	.
food .jane	0.001	91.837	+INF	.
clothes.dick	0.001	18.289	+INF	.
clothes.jane	0.001	24.568	+INF	.

GE with Production (cont'd.)

- ---- VAR y Level of use of a the technology

	LOWER	LEVEL	UPPER	MARGINAL
a1	.	15.714	+INF	.
a2	.	.	+INF	2.846
a3	.	14.286	+INF	.
a4	.	.	+INF	0.600

- Notice the complementarity in the output here versus the output for equation "loss"

GE with Production (cont'd.)

- ---- VAR lambda Marginal utility of wealth

	LOWER	LEVEL	UPPER	MARGINAL
dick	.	0.454	+INF	.
jane	.	0.354	+INF	.

Equilibrium Modeling with MCP

- MCP is a useful approach to the formulation and solution of economic equilibrium problems
  - Some problems can be stated as systems of equations
  - Others need complementarity conditions to complete the formulation
- MCP is also useful for the formulation of certain types of games